

A Turbo Vision ismertetése

II. rész

Az előző számban bemutattuk egy egyszerű keretrendszer létrehozási módját a Turbo Vision segítségével. Ahhoz, hogy megváltoztassuk a státussor tartalmát, a Turbo Vision által kínált alapértelmezett státussort kell felülírnunk. Ennek megvalósítása érdekében az objektumtípus deklarációjában az objektumon belül beírjuk az *InitStatusLine* metódust a következőképpen:

```
TMyApp = Object (TApplication)
  Procedure InitStatusLine; Virtual;
End;
```

Az objektumtípus deklarációját a metódus leírásának kell követnie. A metóduson belül használható függvények közül a legfontosabb a *New*, amely az új státussort definiálja. Felépítése a következő:

```
New (PStatusLine, Init (RectVáltozó,
  NewStatusDef (KezdÉrték, VégÉrték,
  NewStatusDef (Szöveg, kbKonstans, cmKonstans,
  NewStatusDef (Szöveg, kbKonstans, cmKonstans,
  ...)),
  nil)),
  ni),
  ));
```

ahol:

- *PStatusLine* - egy előre definiált *pointer* (mutató);
- *RectVáltozó* - egy *TRect* típusú változó, amely egy téglalap alakú és legalább egy karakter nagyságú területet foglal le a képernyőn;
- *KezdÉrték* - egy 0 és FFFFh (FFFF tizenhatos számrendszerben) közti szám, amely a státussor memóriahelyének lefoglalásához szükséges;
- *VégÉrték* - egy 0 és FFFFh közti szám, amely a státussor memóriahelyének lefoglalásához szükséges (*VégÉrték* > *KezdÉrték*);
- *Szöveg* - a képernyőn megjelent szöveg;
- *kbKonstans* - egy meghatározott billentyűnek megfelelő konstans;
- *cmKonstans* - egy meghatározott parancsnak megfelelő konstans;

A *RectVáltozó* nagyságát a *GetExtent* valamint az *R.A.X*, *R.B.X*, *R.A.Y*, *R.B.Y* eljárásokkal lehet megváltoztatni. A *GetExtent* eljárás maximálisra növeli a téglalapot. Az *R.A.X* és *R.B.X* a téglalap vízszintes, az *R.A.Y* és *R.B.Y* pedig a téglalap függőleges méreteinek a beállításához szükséges. Ha például a státussort a képernyő aljára akarjuk helyezni úgy, hogy egy sor magas legyen és a hossza a képernyő szélességével egyenlő, az *R.A.X* és *R.B.X* eljárásokat nem kell változtatnunk. Ahhoz pedig hogy a sor magassága 1 legyen és a státussor a képernyő aljára kerüljön az *R.A.Y*-nak *R.B.Y-1*-gyel kell egyenlőnek lennie.

A *KezdÉrték* és *VégÉrték* általában 1 ill. \$FFFF, kivétel, ha több státussort definiálunk és ezeket egymás között cserélgetjük. Ilyenkor az FFFF értéket annyi részre osztjuk ahány státussor található, és minden státussor *New* metódusának kezdeti értéke és végértéke a megfelelő intervallumok kezdeti ill. végértékével

lesz egyenlő (pl. két státussor esetében: az első kezdeti értéke 1 végértéke \$8888, a másodiké pedig \$8889 ill. \$FFFF).

A *Szöveg* egy idézőjelek között lévő karaktorsor. Ha ebben a szövegrészben valamit ki akarunk emelni, azt ~ jelek közé tesszük.


A *kbKonstans* a billentyűzetnek megfelelő konstans, mely az adott módon előre értelmezett (pl. *kbAltP* az Alt-P kombinációnak, *kbIns* az *Insert* billentyűnek, *kbF10* az F10 billentyűnek megfelelő konstans).

A *cmKonstans* parancskonstans, amely lehet előreértelmezett is, de definiálhatunk saját konstansokat is.

Most lássuk a programunkat státussorral is ellátva:

```
Program Turbo_Vision_02;  
  
Uses Objects, Drivers, Views, Menus, App;  
  
Type  
  TMyApp = Object (TApplication)  
    Procedure InitStatusLine; Virtual;  
    End;  
Procedure TMyApp.InitStatusLine;  
Var R: TRect;  
Begin  
  GetExtent (R) ;  
  R.A.Y := R.B.Y - 1;  
  StatusLine := New (PStatusLine, Init (R,  
    NewStatusDef (0, $FFFF,  
      NewStatusKey (' ~Alt-X~ Kilépés' , kbAltX, cmQuit,  
        nil),  
        nil)  
    ));  
End;  
  
Var  
  MyApp: TMyApp;  
  
Begin  
  MyApp.Init;  
  MyApp.Run;  
  MyApp.Done;  
End.
```

A program hatására a képernyőn megjelenik:



amely az Alt-X kombinációval zárható be.

Ha egy menüt is létre akarunk hozni, akkor a típusdeklarálásnál kötelező módon meg kell jelennie a következő sornak :

```
Procedure InitMenuBar; Virtual;
```

Egy ilyen menüsor létrehozásához a neki megfelelő metódust is le kell írni, amelynek kötelezően tartalmaznia kell a *New* eljárást. Ennek szintaxisa:

```
New(PMenuBar, Init(RectVáltozó, NewMenu(  
    NewSubMenu(Szoveg, hcKonstans  
       NewItem(Szöveg1, Szöveg2, kbKonstans, cmKonstans,  
                hcKonstans  
       NewItem(Szöveg1, kbKonstans, cmKonstans,  
                hcKonstans  
        ....  
        nil))....,  
    NewSubMenu(Szoveg, hcKonstans  
       NewItem(Szöveg1, Szöveg2, kbKonstans, cmKonstans,  
                hcKonstans  
       NewItem(Szöveg1, kbKonstans, cmKonstans,  
                hcKonstans  
        ....  
        nil))....,  
    .....  
    nil)....,  
));
```

Észrevehető, hogy megjelenik a *hcKonstans* is, amely egy *help* bemenetét jelenti (a *hcNoContext* azt jelenti, hogy nincs help hozzákötve ehhez a parancshoz).

Tehát a menüvel ellátott program a következőképpen mutat:

```
Program Turbo_Vision_03;  
Uses Objects, Drivers, Views, Menus, App;  
Const  
    cmFileOpen = 100;  
    cmNewWin = 101;  
Type  
    TMyApp = Object(TApplication)  
        Procedure InitMenuBar; Virtual;  
        Procedure InitStatusLine; Virtual;  
    End;  
Procedure TMyApp.InitMenuBar;  
Var R: TRect;  
Begin  
    GetExtent(R);  
    R.B.Y := R.A.Y + 1;  
    MenuBar := New(PMenuBar, Init(R, NewMenu(  
        NewSubMenu(' ~F~ile', hcNoContext, NewMenu(  
           NewItem(' ~M~egnyít', ' F3', kbF3, cmFileOpen,  
                    hcNoContext,  
           NewItem(' ~U~j', ' F4', kbF4, cmNewWin,  
                    hcNoContext,  
            NewLine(  
               NewItem(' ~K~ilép', ' Alt-X', kbAltX, cmQuit,  
                        hcNoContext,  
                nil))))) ,
```

```

NewSubMenu(' ~A~blak' , hcNoContext, NewMenu(
  NewItem(' ~K~övetkező' , ' F6' , kbF6, cmNext,
    hcNoContext,
    NewItem(' At~m~éretez' , ' F5' , kbF5, cmZoom,
      hcNoContext,
      nil))),
  nil))
));
End;


Procedure TMyApp.InitStatusLine;
Var R: TRect;
Begin
  GetExtent(R);
  R.A.Y := R.B.Y - 1;
  StatusLine := New(PStatusLine, Init(R,
    NewStatusDef(0, $FFFF,
      NewStatusKey(' ~F10~ Menü' , kbF10, cmMenu,
        NewStatusKey(' ~Alt-X~ Kilép' , kbAltX, cmQuit,
          NewStatusKey(' ~F4~ Uj' , kbF4, cmNewWin,
            NewStatusKey(' ~Alt-F3~ Bezár' , kbAltF3, cmClose,
              nil)))))
    nil)
  ));
End;

Var
  MyApp: TMyApp;

Begin
  MyApp.Init;
  MyApp.Run;
  MyApp.Done;
End.

```

Ennek eredményeképpen a képernyőn a következő keretrendszer jelenik meg:



Dávid K. Zoltán
Kolozsvár